

RとStanで微分方程式 入りのモデル

Tokyo.R #105

伊東宏樹

2023-04-22

自己紹介

名前: 伊東宏樹

勤務先: 森林総合研究所 北海道支所

共訳書:




ロジスティック方程式

生物の個体数の時間変化をあらわす微分方程式

$$\frac{dx}{dt} = rx \left(1 - \frac{x}{K} \right)$$

- x : 個体数
- t : 時間
- r : 内的自然増加率
- K : 環境収容力

模擬データの生成



- 用意したロジスティック曲線にそった値に、対数正規分布にしたがうノイズが加わって点のデータが生成されたとする。
- 微分方程式を組み込んだモデルで、生成された点のデータから、曲線のパラメータ (r , K) を推定する。

Stanで微分方程式

Stanには常微分方程式の数値解を求める関数が組み込まれている。アルゴリズムの異なる関数がいくつか用意されているが、今回は一般的なode_rk45を使用。

```
1 array[] vector ode_rk45(function ode,  
2                       vector initial_state,  
3                       real initial_time,  
4                       array[] real times, ...)
```

引数として与える関数の形式

```
1 vector ode(real time, vector state, ...)
```

Stanモデル: functionsブロック

ロジスティック方程式を定義

```
1 functions {
2   vector logistic(real t, vector x, array[] real theta) {
3     vector[1] dx_dt;
4     real r = theta[1];
5     real K = theta[2];
6
7     dx_dt[1] = r * x[1] * (1 - x[1] / K);
8     return dx_dt;
9   }
10 }
```

data & parametersブロック


```
1 data {
2   int<lower = 0> N;           // number of measurements
3   array[N] real ts;         // measurement times
4   real<lower = 0> y0;        // initial measured value
5   array[N] real<lower = 0> y; // measured values
6 }
7
8 parameters {
9   real<lower = 0> r;         // intrinsic growth rate
10  real<lower = 0> K;         // carrying capacity
11  vector<lower = 0>[1] z0;   // initial value
12  real<lower = 0> sigma;     // noise scale
13 }
```

modelブロック

```
1 model {
2   array[2] real theta = {r, K};
3   array[N] vector[1] z = ode_rk45(logistic, z0, 0, ts,
4                                   theta);
5
6   y0 ~ lognormal(log(z0), sigma);
7   for (n in 1:N)
8     y[n] ~ lognormal(log(z[n]), sigma);
9   // priors
10  r ~ normal(0, 5);
11  K ~ normal(0, 100);
12  z0 ~ normal(0, 100);
13  sigma ~ normal(0, 5);
14 }
```


結果

パラメータの事後平均値をつかってロジスティック曲線を描画(赤線)。



本日の資料

- スライド:

https://ito4303.github.io/stan_ode.html

- リポジトリ:

<https://github.com/ito4303/TokyoR105>

